
Norma Pugh                                           
July 6, 2009 

1 

Integrity Check for the Behavior Enhances Drug Reduction of Incontinence Trial (BE-DRI) Data 
Files 

 
As a partial check of the integrity of the BE-DRI datasets archived in the NIDDK data repository, a set of 
tabulations was performed to verify that published results can be reproduced using the archived datasets. 
Analyses were performed to duplicate results for the data published by Burgio et al [1] in the Annals of 
Internal Medicine in August 2008.  The results of this integrity check are described below.  The full text 
of the Annals of Internal Medicine article can be found in Attachment 1, and the SAS code for our 
tabulations is included in Attachment 2. 
 
The intent of this DSIC is to provide confidence that the data distributed by the NIDDK repository is a 
true copy of the study data.  Our intent is not to assess the integrity of the statistical analyses reported by 
study investigators.  As with all statistical analyses of complex datasets, complete replication of a set of 
statistical results should not be expected on a first (or second) exercise in secondary analysis.  This occurs 
for a number of reasons including differences in the handling of missing data, restrictions on cases 
included in samples for a particular analysis, software coding used to define complex variables, etc.  
Experience suggests that most discrepancies can ordinarily be resolved by consultation with the study 
data coordinating center (DCC), however this process is labor-intensive for both DCC and Repository 
staff.  It is thus not our policy to resolve every discrepancy that is observed in an integrity check.  
Specifically, we do not attempt to resolve minor or inconsequential discrepancies with published results 
or discrepancies that involve complex analyses, unless NIDDK Repository staff suspect that the observed 
discrepancy suggests that the dataset may have been corrupted in storage, transmission, or processing by 
repository staff.  We do, however, document in footnotes to the integrity check those instances in which 
our secondary analyses produced results that were not fully consistent with those reported in the target 
publication. 
 
Background. The Urinary Incontinence Treatment Network (UITN) was set up to establish a group of 
collaborating investigators who will conduct long-term studies, including clinical trials, of the most 
commonly used surgical, pharmacological, and behavioral approaches to the management of urinary 
incontinence in women diagnosed with stress and mixed incontinence. 
  
The UITN consists of nine clinical centers in the U.S. The biostatistical center is located at the New 
England Research Institute in Watertown, Massachusetts. 
  
The Behavior Enhances Drug Reduction of Incontinence (BE-DRI) clinical trial, is a 2-stage randomized 
trial designed to treat women with urge-predominant urinary incontinence. For stage 1, patients were 
randomly assigned to drug therapy alone or drug therapy combined with behavioral training. In stage 2 
(immediately following stage 1), drug therapy was discontinued in both groups [1], [2]. 
 
The Burgio paper aims to determine whether combining antimuscarinic drug therapy with supervised 
behavioral training, compared with drug therapy alone, improves the ability of women with urge 
incontinence to achieve clinically important reductions in incontinence episodes and to sustain these 
improvements after discontinuing drug therapy [1]. 

Participant Characteristics. Table 1 in the publication [1] reports on demographic and clinical 
characteristics of women enrolled in the BE-DRI Trial, by treatment.  Table A lists the variables we used 
in our replication. 
 

Table A: Variables Used to Replicate Table 1 
 

Table Variable Variables Used in Replication 
Treatment group Dataset bd_aid: assignment, Form F211: assignment 
Age Dataset bd_aid: age 
Ethnicity Form F201: ethnicity, race_wh, race_bl, race_ai 
Marital status Form F201: marital 
Level of education Form F201: edu 
Nam-Powers occupational status score Form F201: occup_code (if value>0), else occup2_code 
Duration of urge incontinence Dataset bd_aid: age , Form F201: age_began 
Previous nonsurgical treatment for 
incontinence Form F202: ui_treat 

Previous incontinence surgery Form F202: ui_tx_surg 
Diuretic use Form F263: diuretic, where visit=‘RAND’ 
Body mass index Form F204: bmi 
Menopausal status Form F202: meno 
Type and frequency of incontinence Form F209: ic_type, ic_freq 

 
In Table B, we compare the results calculated from the archived datasets to the results published in Table 
1, Participant Characteristics by Treatment Group.  As Table B shows, the results are similar.

Table B: Comparison of Values Computed in Integrity Check to Reference Article Table 1 Values, 
Treatment = Combination Therapy 

 
Characteristic Burgio Integrity Check Diff 
Sample size 154 154 0 
Mean age (SD), y 55.8 (14.2) 55.8 (14.2) 0 
Ethnicity, n (%)    
  Hispanic 13 (9) 13 (8) 0 (1) 
  Non-Hispanic white 105 (69) 106 (69) 1 (0) 
  Non-Hispanic black 22 (14) 22 (14) 0 
  Other non-Hispanic 13 (9) 13 (8) 0 (1) 
Marital status, n (%)    
   Married or living with partner 70 (46) 70 (45) 0 (1) 
   Not married 84 (54) 84 (55) 0 (1) 
Level of education, n (%)    
   High school or less 40 (26) 40 (26) 0 
   Some college or associate’s degree 58 (38) 58 (38) 0 
   College 32 (21) 32 (21) 0 
   Graduate or professional degree 24 (16) 24 (16) 0 
Mean Nam-Powers occupational status score (SD) 57.5 (23.3) 57.5 (23.3) 0 
Mean duration of urge incontinence (SD), y 9.8 (9.5) 9.8 (9.5) 0 
Previous nonsurgical treatment for incontinence, n (%) 59 (38) 59 (38) 0 
Previous incontinence surgery, n (%) 19 (12) 19 (12) 0 
Diuretic use, n (%) 15 (10) 15 (10) 0 
Mean body mass index (SD), kg/m2 33.2 (9.5) 33.2 (9.5) 0 
Menopausal status, n (%)    
   Premenopausal 35 (23) 35 (23) 0 
   Postmenopausal 96 (62) 96 (62) 0 
   Perimenopausal 20 (13) 
   Unsure 3 (2) 23 (15) 0* 

Type and frequency of incontinence, n (%)    
   Urge, 7-13 episodes/wk 2 (1.3) 2 (1.3) 0 
   Urge, ≥ 14 episodes/wk 2 (1.3) 2 (1.3) 0 
   Mixed, 7-13 episodes/wk 46 (29.9) 46 (29.9) 0 
   Mixed, ≥ 14 episodes/wk 104 (67.5) 104 (67.5) 0 
 
*For purposes of dataset de-identification, ’Menopausal status’ values of ‘Perimenopausal’ and ‘Unsure’ 
are combined in the archived data files.

Table B: Comparison of Values Computed in Integrity Check to Reference Article Table 1 Values, 

Treatment = Drug Therapy Alone 
 

Characteristic Burgio Integrity Check Diff 
Sample size 153 153 0 
Mean age (SD), y 58.0 (13.5) 58.0 (13.5) 0 
Ethnicity, n (%)    
  Hispanic 17 (11) 17 (11) 0 
  Non-Hispanic white 85 (56) 86 (56) 1 (0) 
  Non-Hispanic black 35 (23) 35 (23) 0 
  Other non-Hispanic 15 (10) 15 (10) 0 
Marital status, n (%)    
   Married or living with partner 68 (44) 68 (44) 0 
   Not married 85 (56) 85 (56) 0 
Level of education, n (%)    
   High school or less 31 (20) 31 (20) 0 
   Some college or associate’s degree 60 (39) 60 (39) 0 
   College 40 (26) 40 (26) 0 
   Graduate or professional degree 22 (14) 22 (14) 0 
Mean Nam-Powers occupational status score (SD) 64.0 (24.3) 64.0 (24.3) 0 
Mean duration of urge incontinence (SD), y 9.1 (10.3) 9.1 (10.3) 0 
Previous nonsurgical treatment for incontinence, n (%) 46 (30) 46 (30) 0 
Previous incontinence surgery, n (%) 22 (14) 22 (14) 0 
Diuretic use, n (%) 14 (9) 14 (9) 0 
Mean body mass index (SD), kg/m2 32.3 (7.6) 32.3 (7.6) 0 
Menopausal status, n (%)    
   Premenopausal 31 (20) 31 (20) 0 
   Postmenopausal 102 (67) 102 (67) 0 
   Perimenopausal 18 (12) 
   Unsure 2 (1) 20 (13) 0* 

Type and frequency of incontinence, n (%)    
   Urge, 7-13 episodes/wk 2 (1.3) 2 (1.3) 0 
   Urge, ≥ 14 episodes/wk 4 (2.6) 4 (2.6) 0 
   Mixed, 7-13 episodes/wk 46 (30.1) 46 (30.1) 0 
   Mixed, ≥ 14 episodes/wk 101 (66.0) 101 (66.0) 0 
 
*For purposes of dataset de-identification, ’Menopausal status’ values of ‘Perimenopausal’ and ‘Unsure’ 
are combined in the archived data files.

Participant Outcomes.  Table C lists the variables we used in our replication of selected outcomes 
presented in both the publication text and tables [1]. 
 

Table C: Variables Used to Replicate Selected Outcomes 
 

Text/Table Variable Variables Used in Replication 
Treatment group Dataset bd_aid: assignment, Form F211: assignment 

‘Completely satisfied’ with progress at 
end of stage 1 

Form F261, where visit=‘VS08’; Form F263, where 
visit=‘VS08’; Form F266, where visit=‘VS08’; Form 
F267: prog_satis, where visit=‘VS05’ 

‘Completely satisfied’ with progress at 8 
months 

Form F261, where visit=‘VS08’; Form F263, where 
visit=‘VS08’; Form F266, where visit=‘VS08’; Form 
F267: prog_satis, where visit=‘VS08’ 

Improvement is ‘better’ or ‘much better’ at 
end of stage 1 

Form F261, where visit=‘VS08’; Form F263, where 
visit=‘VS08’; Form F266, where visit=‘VS08’; Form 
F267: over_satis, where visit=‘VS05’ 

Improvement is ‘better’ or ‘much better’ at 
8 months 

Form F261, where visit=‘VS08’; Form F263, where 
visit=‘VS08’; Form F266, where visit=‘VS08’; Form 
F267: over_satis, where visit=‘VS08’ 

Treatment failure: resumed drug use or 
other therapy before 8-month visit 

Form F263: problem_drug, where visit < ‘VS08’; Form 
F261: new_urge, where visit < ‘VS08’ 

Treatment failure: resumed drug use 
before 8-month visit Form F263: problem_drug, where visit < ‘VS08’ 

Treatment failure: resumed other treatment 
before 8-month visit Form F261: new_urge, where visit < ‘VS08’ 

 
In Table D, we compare the results calculated from the archived datasets to the published results.  As 
Table D shows, the results are similar.

Table D: Comparison of Values Computed in Integrity Check to Reference Article Text and Table 
Values, Treatment = Combination Therapy 

 
Characteristic Burgio Integrity Check Diff 
‘Completely satisfied’ with progress at end of stage 1 53% 51% 2 
‘Completely satisfied’ with progress at 8 months 33% 32% 1 
Improvement is ‘better’ or ‘much better’ at end of stage 1 90% 89% 1 
Improvement is ‘better’ or ‘much better’ at 8 months 69% 68% 1 
Treatment failure: resumed drug use or other therapy before 
8-month visit N=29 N=31 2 

Treatment failure: resumed drug use before 8-month visit N=29 N=31 2 
Treatment failure: resumed other treatment before 8-month 
visit N=26 N=26 0 

 
Treatment = Drug Therapy Alone 

 
Characteristic Burgio Integrity Check Diff 
‘Completely satisfied’ with progress at end of stage 1 40% 41% 1 
‘Completely satisfied’ with progress at 8 months 20% 21% 1 
Improvement is ‘better’ or ‘much better’ at end of stage 1 77% 73% 4 
Improvement is ‘better’ or ‘much better’ at 8 months 43% 44% 1 
Treatment failure: resumed drug use or other therapy before 
8-month visit N=36 N=36 0 

Treatment failure: resumed drug use before 8-month visit N=34 N=34 0 
Treatment failure: resumed other treatment before 8-month 
visit N=34 N=34 0 

ATTACHMENT 1 
 

Full Text of Article 
Kathryn L. Burgio, PhD, et al; Behavioral Therapy to Enable Women with Urge 
Incontinence to Discontinue Drug Treatment; Annals of Internal Medicine; 5 August 2008; 
Volume 149, Number 3; pages 161-169, W-35 - W-37. 

NOTE. Single copies of articles published in scientific journals are included with this documentation. 
These articles are copyrighted, and the repository has purchased ONE reprint from their publisher to 
include with this documentation. If additional copies are made of these copyrighted articles, users are 
advised that payment is due to the copyright holder (typically the publisher of the scientific journal).  
 
 



ATTACHMENT 2 
 

SAS Code for Tabulations from the BE-DRI Datasets in the NIDDK Repository 

 
 
 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\BE-DRI\PrimaryPaper\table1.sas 
/* Author:  Norma Pugh 
/* Date:    03 July 2009 
/* Purpose: Replicate table 1 results. 
/* 
/***************************************************************************************/ 
/* DATA SOURCES */ 
libname sasdb 'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery\DataSets'; 
libname fmts  'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery'; 
 
 
/* FORMATS */ 
options fmtsearch=(fmts.formats) nofmterr mprint; 
proc format; 
 value yn 1='1=Yes' 
          2='2=No'; 
 value meno 1='1=PreMeno' 
            2='2=PostMeno' 
            3='3=PeriMeno' 
            4='4=Unsure'; 
 value trt 1='DrugOnly (treat=1)' 
           2='Combo (treat=2)'; 
run; 
 
 
/*******************/ 
/* STUDY VARIABLES */ 
/*******************/ 
data treatment; 
 merge sasdb.f211(keep=aid assignment rename=(assignment=treat211)) 
       sasdb.bd_aid(keep=aid assignment rename=(assignment=treatbd)); 
 by aid; 
 if treat211>. then treat=treat211; else treat=treatbd; 
run;        
 
data table1; 
 merge sasdb.bd_aid(keep=aid age) 
       sasdb.f201(keep=aid ethnicity race_wh race_bl race_ai marital edu occup_code 
occup2_code age_began) 
       sasdb.f202(keep=aid ui_treat ui_tx_surg meno) 
       sasdb.f204(keep=aid bmi) 
       sasdb.f209(keep=aid ic_type ic_freq) 
       sasdb.f263(keep=aid visit diuretic where=(visit='RAND')) 
       treatment(keep=aid treat); 
 by aid; 
 
 /* Ethnicity */ 
 hispanic=ethnicity; 
  
 if ethnicity=2 then do; 
  if race_wh=1 & race_bl=2 & race_ai=2 then race='1_White'; 
  else if race_wh=2 & race_bl=1 & race_ai=2 then race='2_Black'; 
  else race='3_Other'; 
 end; 
 label hispanic='Hispanic?' 



       race='Racial group'; 
 
 /* Marital status */ 
 if marital=1 then mar_stat='1_married/cohab'; 
 else if marital in(2,3,4,5,6) then mar_stat='2_notmarried'; 
 label mar_stat='Re-coded marital status'; 
 
 /* Level of education */ 
 if edu=1 then educate='1_leHS'; 
 else if edu=2 then educate='2_AD'; 
 else if edu=3 then educate='3_BD'; 
 else if edu=4 then educate='4_GD'; 
 label educate='Re-coded education level'; 
 
 /* Nam-Powers occupational status score */ 
 if occup_code>0 then NAM=occup_code; 
  else if occup2_code>0 then NAM=occup2_code; 
  else NAM=.; 
 label NAM='Re-coded Nam-Powers';  
  
 /* Duration of urge incontinence */ 
 duratn_urge=age-age_began; 
 label duratn_urge='Duration of urge incontenence (y)'; 
run; 
 
 
/******************/ 
/* GET STATISTICS */ 
/******************/ 
proc freq data=table1 noprint; tables treat / out=denom; run; 
data denom(keep=treat count rename=(count=denom)); set denom; run; 
 
%macro frq(var,fmt); 
 proc freq data=table1 noprint; 
  tables treat*&var / out=frqstats(drop=percent); 
  &fmt; 
 run; 
  
 data frqstats; 
  merge frqstats denom; 
  by treat; 
  pct=(count/denom)*100; 
 run; 
   
 proc print data=frqstats; title"Frequency Counts: &var"; format treat trt.; run; 
%mend frq; 
 
 
%frq(HISPANIC,%str(format hispanic yn.)); 
%frq(RACE,); 
%frq(MAR_STAT,); 
%frq(EDUCATE); 
%frq(UI_TREAT,%str(format ui_treat yn.)); 
%frq(UI_TX_SURG,%str(format ui_tx_surg yn.)); 
%frq(DIURETIC,%str(format diuretic yn.)); 
/* NOTE: To de-identify data, DCC collapsed meno categories 3 & 4 */ 
%frq(MENO,%str(format meno meno.)); 



%frq(IC_TYPE*IC_FREQ); 
 
 
proc sort data=table1; by treat; run; 
proc means data=table1 n mean stddev min max; by treat; format treat trt.; 
 var age nam bmi duratn_urge; 
 title'Summary of Means'; 
run; 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\BE-DRI\PrimaryPaper\table3.sas 
/* Author:  Norma Pugh 
/* Date:    03 July 2009 
/* Purpose: Replicate selected table 3 results. 
/* 
/***************************************************************************************/ 
/* DATA SOURCES */ 
libname sasdb 'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery\DataSets'; 
libname fmts  'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery'; 
 
 
/* FORMATS */ 
options fmtsearch=(fmts.formats) nofmterr mprint; 
proc format; 
 value trt 1='DrugOnly (treat=1)' 
           2='Combo (treat=2)'; 
run; 
 
 
/*******************/ 
/* STUDY VARIABLES */ 
/*******************/ 
/* Treatment assignment */ 
data treatment(keep=aid treat); 
 merge sasdb.f211(keep=aid assignment rename=(assignment=treat211)) 
       sasdb.bd_aid(keep=aid assignment rename=(assignment=treatbd)); 
 by aid; 
 if treat211>. then treat=treat211; else treat=treatbd; 
run;    
 
 
/* Count Macro */ 
%macro count(out,in,var,visit,title); 
 data &out; 
  set &in(keep=aid visit &var where=(&visit)); 
  if &var=1; 
  keep aid; 
 run; 
 
 proc sort data=&out nodup; by aid; run; 
 
 data &out; merge &out(in=x1) treatment(in=x2); by aid; if x1; run; 
 
 title"&title"; 
 proc freq data=&out order=formatted; tables treat / list missing nopct nocum; format 
treat trt.; run; 
%mend count; 
 
 
/* Subjects requested to resume drug or other therapy before 8-month visit */ 
%count(resume_drg,sasdb.f263,problem_drug,%str(visit<'VS08'),%str(Resumed Drug Tx Before 
8-month Visit)); 
%count(resume_oth,sasdb.f261,new_urge,%str(visit<'VS08'),%str(Resumed Other Tx Before 8-
month Visit)); 
 



/* # Distinct subjects who resumed drug or other therapy before 8 months */ 
data resume_all; merge resume_drg resume_oth; by aid; run; 
proc freq data=resume_all order=formatted; tables treat / list missing nopct nocum; 
format treat trt.; 
 title'Resumed Drug or Other Tx Before 8-month Visit'; 
run; 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\BE-DRI\PrimaryPaper\ratings.sas 
/* Author:  Norma Pugh 
/* Date:    03 July 2009 
/* Purpose: Replicate selected patient ratings. 
/* 
/***************************************************************************************/ 
/* DATA SOURCES */ 
libname sasdb 'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery\DataSets'; 
libname fmts  'R:\03_Data_And_Tools\Studies\UITN-BeDri\DCC-Delivery'; 
 
 
/* FORMATS */ 
options fmtsearch=(fmts.formats) nofmterr mprint; 
proc format; 
 value trt 1='DrugOnly (treat=1)' 
           2='Combo (treat=2)'; 
run; 
 
 
/*******************/ 
/* STUDY VARIABLES */ 
/*******************/ 
/* Treatment assignment */ 
data treatment(keep=aid treat); 
 merge sasdb.f211(keep=aid assignment rename=(assignment=treat211)) 
       sasdb.bd_aid(keep=aid assignment rename=(assignment=treatbd)); 
 by aid; 
 if treat211>. then treat=treat211; else treat=treatbd; 
run;    
 
/* # subjects with complete follow-up */ 
data comp_m8; 
 merge sasdb.f261(keep=aid visit where=(visit='VS08') in=x1) 
      sasdb.f263(keep=aid visit where=(visit='VS08') in=x2) 
      sasdb.f266(keep=aid visit where=(visit='VS08') in=x3) 
      treatment(in=x4); 
 by aid; 
 if x1 & x2 & x3 & x4; 
run; 
proc freq order=formatted; tables treat / list missing; format treat trt.; run; 
 
 
/* Patient Ratings */ 
data ratings_stg1; merge sasdb.f267(where=(visit_num='VS05') in=x1) treatment(in=x2) 
comp_m8(in=x3); by aid; if x1 & x3; run; 
data ratings_m8; merge sasdb.f267(where=(visit_num='VS08') in=x1) treatment(in=x2) 
comp_m8(in=x3); by aid; if x1 & x3; run; 
 
/* Counts */ 
title'Progress at end of stage 1 (rating=1)'; 
proc freq data=ratings_stg1 order=formatted; tables treat*prog_satis / list missing; 
format treat trt.; run; 
title'Progress at month 8 (rating=1)'; 
proc freq data=ratings_m8 order=formatted; tables treat*prog_satis / list missing; format 
treat trt.; run; 



 
title'Improvement with treatment at end of stage 1 (rating=1 or 2)'; 
proc freq data=ratings_stg1 order=formatted; tables treat*over_satis / list missing; 
format treat trt.; run; 
title'Improvement with treatment at month 8 (rating=1 or 2)'; 
proc freq data=ratings_m8 order=formatted; tables treat*over_satis / list missing; format 
treat trt.; run; 

